DIGITAL EQUIPMENT CORPORATION

Resource Handbook $.50

education

HUNTINGTON II Simulation Program—SLITS



computers are for kids — digital — EduSystems—expandable, economical

DIGITAL EQUIPMENT CORPORATION
EDUCATIONAL PUBLICATIONS

A partial list of the publications in the continuing series of curriculum material published by Digital for use with EduSystems and RSTS are listed below. Please inquire directly for prices on classroom quantities.

Additional publications may be obtained from:

Software Distribution Center
Digital Equipment Corporation
Maynard, Massachusetts 01754

Populution: Self teaching BASIC Workbook	$2.00
BASIC Matrix Operations, Project Solo	$1.00
Problems for Computer Mathematics	$1.25
Advanced Problems for Computer Mathematics	2.00
BASIC Application Programs	
Mathematics I	$1.00
Mathematics II	1.00
Science	1.00
Business and Social Studies	1.00
Plotting	1.00
Huntington I Application Programs	
MATHEMATICS	$2.00
TEACHER ASSISTANCE	1.00
Huntington I Simulation Programs	
BIOLOGY	$1.00
CHEMISTRY	2.00
EARTH SCIENCE	1.00
PHYSICS	2.00
SOCIAL STUDIES	1.00
Huntington II Simulation Modules	
Student Workbook	$0.30
Teacher's Guide	.30
Resource Handbook	.50
Program Paper Tape	1.00



SLITS

A COMPUTER SIMULATION of YOUNG'S DOUBLE-SLIT EXPERIMENT

RESOURCE HANDBOOK

Developed and programmed by:

Anthony Caggiano
Patchogue High School
Patchogue, New York

Teacher support material written by:

Elizabeth A. Williamson
Northwest Regional Educational Laboratory
Portland, Oregon

HUNTINGTON TWO COMPUTER PROJECT

15 January 1971

The work of the Huntington Two Computer Project is partially
supported by the National Science Foundation, Grant GW-5883.

RESOURCE MATERIAL

SLITS

Table of Contents

COVER PHOTO

A continuous argon laser is being used by engineers of the Goodard Space Flight Center, Greenbelt, Maryland in experiments to send a message by laser to a satellite in orbit. This is an advanced application of monochromatic light used in young's light interference experiments. Photo courtesy of NASA.

I. HISTORICAL ANECDOTES

When Young originally did his experiment in 1801, his apparatus was somewhat different than that which is used today. He used sunlight passing through a pinhole as a point source of light. The light spreading out from this pinhole source fell upon an opaque barrier which contained two pinholes placed very close together and located equidistant from the source. Light originating from the pinhole source at any moment passed through the other two pinholes at the same time; the light at these two pinholes was then always in phase. With such an arrangement, the interference pattern of the light emerging from the pair of pinholes did not shift; it could be observed.

Reference: PSSC, *PHYSICS*, p.297.

II. SAMPLE RUNS

The sample runs included in this section are intended to be indicative of the output format and the degree of flexibility available in the program. They also are useful for checking the program when it first is entered into the computer to ensure proper operation.

YOUNG'S DOUBLE SLIT EXPERIMENT

L = 2 METERS   W = 6000 ANGSTROMS   D = .5 MILLIMETERS

```
DISTANCE (MM'S FROM CENTER)
        -.26   .                  *
        -.24   .                   *
        -.22   .                  *
        -.2    .              *
        -.18   .         *
        -.16   .    *
        -.14   .*
        -.12   *
        -.1    .*
        -.08   .    *
        -.06   .         *
        -.04   .              *
        -.02   .                  *
...............0....INTENSITY......*
        .02    .                  *
        .04    .              *
        .06    .         *
        .08    .    *
        .1     .*
        .12    *
        .14    .*
        .16    .    *
        .18    .         *
        .2     .              *
        .22    .                  *
        .24    .                   *
        .26    .                  *
```

ABOVE IS AN ILLUSTRATIVE RUN WITH PRE-DETERMINED VALUES FOR WAVELENGTH (W), DISTANCE BETWEEN SLITS AND SCREEN (L), AND SLIT SEPARATION - CENTER TO CENTER (D). NOW YOU MAY VARY THESE PARAMETERS, ONE AT A TIME.

*****

```
WHAT IS THE NEW SLIT SEPARATION (D) IN MILLIMETERS? 1

L = 2 METERS    W = 6000 ANGSTROMS    D = 1 MILLIMETERS

DISTANCE (MM'S FROM CENTER)
       -.26   .             *
       -.24   .                  *
       -.22   .             *
       -.2    .    *
       -.18   *
       -.16   .    *
       -.14   .             *
       -.12   .                  *
       -.1    .             *
       -.08   .    *
       -.06   *
       -.04   .    *
       -.02   .             *
..............O....INTENSITY.....*
        .02   .             *
        .04   .    *
        .06   *
        .08   .    *
        .1    .             *
        .12   .                  *
        .14   .             *
        .16   .    *
        .18   *
        .2    .    *
        .22   .             *
        .24   .                  *
        .26   .             *

WOULD YOU LIKE TO TRY ANOTHER VALUE OF D (1-YES, 0-NO)? 1
WHAT IS THE NEW SLIT SEPARATION (D) IN MILLIMETERS? .25
```

```
L = 2 METERS    W = 6000 ANGSTROMS    D = .25 MILLIMETERS

DISTANCE (MM'S FROM CENTER)
       -.26   *
       -.24   *
       -.22   *
       -.2    .*
       -.18   .  *
       -.16   .    *
       -.14   .      *
       -.12   .         *
       -.1    .            *
       -.08   .              *
       -.06   .                *
       -.04   .                  *
       -.02   .                   *
..............O....INTENSITY......*
        .02   .                   *
        .04   .                  *
        .06   .                *
        .08   .              *
        .1    .            *
        .12   .         *
        .14   .      *
        .16   .    *
        .18   .  *
        .2    .*
        .22   *
        .24   *
        .26   *

WOULD YOU LIKE TO TRY ANOTHER VALUE OF D (1-YES, 0-NO)? 0

*****

WHAT IS THE NEW WAVELENGTH (W) IN ANGSTROMS? 3000
```

L = 2 METERS   W = 3000 ANGSTROMS   D = .5 MILLIMETERS

```
DISTANCE (MM'S FROM CENTER)
        -.26  .              *
        -.24  .                   *
        -.22  .              *
        -.2   .    *
        -.18  *
        -.16  .    *
        -.14  .              *
        -.12  .                   *
        -.1   .              *
        -.08  .    *
        -.06  *
        -.04  .    *
        -.02  .              *
..............O....INTENSITY......*
         .02  .              *
         .04  .    *
         .06  *
         .08  .    *
         .1   .              *
         .12  .                   *
         .14  .              *
         .16  .    *
         .18  *
         .2   .    *
         .22  .              *
         .24  .                   *
         .26  .              *
```

WOULD YOU LIKE TO TRY ANOTHER VALUE OF W (1-YES, 0-NO)? 1
WHAT IS THE NEW WAVELENGTH (W) IN ANGSTROMS? 15000
A WAVELENGTH OF 15000 IS INFRARED LIGHT AND NOT VISIBLE.
THE INTERFERENCE PATTERN WILL BE VISIBLE USING DETECTORS
ONLY. TRY ANOTHER WAVELENGTH.
WHAT IS THE NEW WAVELENGTH (W) IN ANGSTROMS? 6900

```
L = 2 METERS     W = 6900 ANGSTROMS     D = .5 MILLIMETERS

DISTANCE (MM'S FROM CENTER)
        -.26  .                  *
        -.24  .                *
        -.22  .            *
        -.2   .       *
        -.18  .   *
        -.16  .*
        -.14  *
        -.12  .*
        -.1   .   *
        -.08  .       *
        -.06  .           *
        -.04  .               *
        -.02  .                  *
..............O....INTENSITY......*
         .02  .                  *
         .04  .               *
         .06  .           *
         .08  .       *
         .1   .   *
         .12  .*
         .14  *
         .16  .*
         .18  .   *
         .2   .       *
         .22  .            *
         .24  .                *
         .26  .                  *

WOULD YOU LIKE TO TRY ANOTHER VALUE OF W (1-YES, 0-NO)? 0

*****

WHAT IS THE NEW DISTANCE FROM SLITS TO SCREEN (L) IN METERS? 5
```

```
L = 5 METERS    W = 6000 ANGSTROMS    D = .5 MILLIMETERS

DISTANCE (MM'S FROM CENTER)
        -.26   .*
        -.24   . *
        -.22   .  *
        -.2    .    *
        -.18   .      *
        -.16   .        *
        -.14   .          *
        -.12   .            *
        -.1    .              *
        -.08   .                *
        -.06   .                 *
        -.04   .                  *
        -.02   .                   *
...............O....INTENSITY......*
         .02   .                   *
         .04   .                  *
         .06   .                 *
         .08   .                *
         .1    .              *
         .12   .            *
         .14   .          *
         .16   .        *
         .18   .      *
         .2    .    *
         .22   .  *
         .24   . *
         .26   .*

WOULD YOU LIKE TO TRY ANOTHER VALUE OF L (1-YES, 0-NO)? 0

*****

YOU WILL NOW BE GIVEN A LIGHT SOURCE OF UNKNOWN
WAVELENGTH.  YOU WILL SPECIFY THE SLIT SEPARATION (D),
AND THE DISTANCE FROM SLITS TO SCREEN (L).
WHAT IS THE NEW SLIT SEPARATION (D) IN MILLIMETERS? .5
WHAT IS THE NEW DISTANCE FROM SLITS TO SCREEN (L) IN METERS? 4
```

```
L = 4 METERS     W = ? ANGSTROMS    D = .5 MILLIMETERS

DISTANCE (MM'S FROM CENTER)
        -.26   *
        -.24   *
        -.22   *
        -.2    .*
        -.18   .  *
        -.16   .    *
        -.14   .       *
        -.12   .          *
        -.1    .             *
        -.08   .               *
        -.06   .                 *
        -.04   .                   *
        -.02   .                    *
...............0....INTENSITY.......*
         .02   .                    *
         .04   .                   *
         .06   .                 *
         .08   .               *
         .1    .             *
         .12   .          *
         .14   .       *
         .16   .    *
         .18   .  *
         .2    .*
         .22   *
         .24   *
         .26   *

WOULD YOU LIKE A PLOT FOR OTHER VALUES OF D AND L (1-YES, 0-NO)? 0
WHAT DO YOU THINK THE UNKNOWN WAVELENGTH (W) IS? 6000
PRETTY GOOD!  THE WAVELENGTH WAS  6000  ANGSTROMS.
WOULD YOU LIKE TO TRY ANOTHER UNKNOWN WAVELENGTH(1-YES, 0-NO)? 0

*****
```

```
WOULD YOU LIKE A PLOT WITH YOUR OWN VALUES FOR WAVELENGTH
 (W), SLIT SEPARATION (D), AND DISTANCE FROM SLITS TO
SCREEN (L) (1-YES, 0-NO)? 1
WHAT IS THE NEW WAVELENGTH (W) IN ANGSTROMS? 5500
WHAT IS THE NEW SLIT SEPARATION (D) IN MILLIMETERS? .75
WHAT IS THE NEW DISTANCE FROM SLITS TO SCREEN (L) IN METERS? 3

L = 3 METERS    W = 5500 ANGSTROMS    D = .75 MILLIMETERS

DISTANCE (MM'S FROM CENTER)
        -.26   .              *
        -.24   .                  *
        -.22   .                    *
        -.2    .                  *
        -.18   .              *
        -.16   .         *
        -.14   .   *
        -.12   *
        -.1    *
        -.08   .   *
        -.06   .         *
        -.04   .              *
        -.02   .                  *
...............0....INTENSITY.......*
         .02   .                  *
         .04   .              *
         .06   .         *
         .08   .   *
         .1    *
         .12   *
         .14   .   *
         .16   .         *
         .18   .              *
         .2    .                  *
         .22   .                    *
         .24   .                  *
         .26   .              *

ANOTHER ONE (1-YES, 0-NO)
? 0

**********.

HOPE YOU HAD FUN!
```

III. PROGRAM LISTING

A listing of the program is included on the following pages. It is written in the version of BASIC which existed on the Digital Equipment Corporation TSS/8 in August 1970.

The program, as listed, should operate in any machine with a BASIC compiler with the implementation of BASIC described in *BASIC: AN INTRODUCTION TO COMPUTER PROGRAMMING USING THE BASIC LANGUAGE*, W. F. Sharpe, The Free Press; with the exceptions that the TAB function is used for generation of the output plots, and RANDOMIZE is used to generate random wavelengths to test the student's grasp of the concepts related to this experiment. If TAB is not available, some other plotting routine must be substituted for the one included here. If RANDOMIZE is not available, some other method of selecting randomly the unknown wavelength must be substituted.

III. PROGRAM LISTING

```
50 REM SLITS--COPYRIGHT 1971, STATE UNIVERSITY OF NEW YORK
51 REM LATEST REVISION 8-27-72
100 REM YOUNG'S DOUBLE SLIT EXPERIMENT
101 REM A.C. CAGGIANO
102 REM REVISED 7/28/70 (L. BRAUN, D. PESSEL)
103 REM IMPORTANT VARIABLES:  L-DISTANCE BETWEEN SLITS+SCREEN;
104 REM W-WAVELENGTH; D-SLIT SEPARATION(CENTER TO CENTER)
105 REM
106 REM U:  PRINT PARAMETER FOR UNKNOWN WAVELENGTH
107 LET U=0
110 PRINT " ", "YOUNG'S DOUBLE SLIT EXPERIMENT"
111 PRINT
120 REM ILLUSTRATIVE RUN
130 LET L=2
140 LET W=6000
150 LET D=.5
160 REM PLOT ROUTINE
170 GOSUB 855
171 PRINT
180 PRINT "ABOVE IS AN ILLUSTRATIVE RUN WITH PRE-DETERMINED"
181 PRINT "VALUES FOR WAVELENGTH (W), DISTANCE BETWEEN SLITS"
182 PRINT "AND SCREEN (L), AND SLIT SEPARATION - CENTER TO"
183 PRINT "CENTER (D).  NOW YOU MAY VARY THESE PARAMETERS,"
184 PRINT "ONE AT A TIME."
186 PRINT
187 PRINT "*****"
188 PRINT
190 REM D INPUT SUBROUTINE
200 GOSUB 922
210 REM PLOT ROUTINE
220 GOSUB 855
221 PRINT
230 PRINT "WOULD YOU LIKE TO TRY ANOTHER VALUE OF D (1-YES, 0-NO)";
240 INPUT Q1
250 IF Q1>0 THEN 200
260 PRINT
261 PRINT "*****"
262 PRINT
270 REM RESET D
280 LET D=.5
290 REM W INPUT SUBROUTINE
300 GOSUB 944
310 REM PLOT SUBROUTINE
320 GOSUB 855
321 PRINT
330 PRINT "WOULD YOU LIKE TO TRY ANOTHER VALUE OF W (1-YES, 0-NO)";
340 INPUT Q2
350 IF Q2>0 THEN 300
360 PRINT
```

```
361 PRINT "*****"
362 PRINT
370 REM RESET W
380 LET W=6000
390 REM L INPUT SUBROUTINE
400 GOSUB 902
410 REM PLOT SUBROUTINE
420 GOSUB 855
421 PRINT
430 PRINT "WOULD YOU LIKE TO TRY ANOTHER VALUE OF L (1-YES, 0-NO)";
440 INPUT Q3
450 IF Q3>0 THEN 400
460 PRINT
461 PRINT "*****"
462 PRINT
470 REM RESET L
480 LET L=2
490 PRINT "YOU WILL NOW BE GIVEN A LIGHT SOURCE OF UNKNOWN"
491 PRINT "WAVELENGTH.  YOU WILL SPECIFY THE SLIT SEPARATION (D),"
492 PRINT "AND THE DISTANCE FROM SLITS TO SCREEN (L)."
507 REM Q5 DETERMINES IF W IS TO BE CHANGED
508 LET Q5=0
520 REM D INPUT SUBROUTINE
530 GOSUB 922
550 REM L INPUT SUBROUTINE
560 GOSUB 902
565 REM CHANGE W?
566 IF Q5>0 THEN 601
570 REM RANDOMLY DETERMINE WAVELENGTH
580 RANDOMIZE
590 LET W=1000*INT(3*RND(X)+4.5)
600 REM PLOT SUBROUTINE (UNKNOWN W)
601 LET U=1
605 GOSUB 855
606 PRINT
610 PRINT "WOULD YOU LIKE A PLOT FOR OTHER VALUES OF D AND L ";
611 PRINT "(1-YES, 0-NO)";
620 INPUT Q5
630 IF Q5>0 THEN 530
640 PRINT "WHAT DO YOU THINK THE UNKNOWN WAVELENGTH (W) IS";
650 INPUT W1
660 IF ABS(W1-W)<.1*W THEN 700
670 PRINT "YOU ARE MORE THEN 10% OFF.  TO HELP YOU, YOU MAY ";
680 PRINT "OBTAIN MORE PLOTS."
690 GO TO 610
700 PRINT "PRETTY GOOD!  THE WAVELENGTH WAS "W" ANGSTROMS."
701 PRINT "WOULD YOU LIKE TO TRY ANOTHER UNKNOWN WAVELENGTH";
702 PRINT "(1-YES, 0-NO)";
703 INPUT Q6
704 IF Q6<1 THEN 967
705 PRINT "YOU MAY SPECIFY A NEW SLIT SEPARATION (D) AND DISTANCE"
706 PRINT "FROM SLITS TO SCREEN (L)."
707 GO TO 508
```

```
849 REM
850 REM PLOT ROUTINE
855 PRINT
856 PRINT
857 REM U>0 DO NOT PRINT WAVELENGTH
858 IF U>0 THEN 870
860 PRINT "L ="L"METERS    W ="W"ANGSTROMS    D ="D"MILLIMETERS"
861 PRINT
865 GO TO 875
870 PRINT "L ="L"METERS    W = ? ANGSTROMS    D ="D"MILLIMETERS"
871 PRINT
875 PRINT "DISTANCE (MM'S FROM CENTER)"
880 REM A:PLOT LOWER LIMIT (MM'S); B:UPPER LIMIT (MM'S)
881 LET A=-.26
882 LET B=.26
883 REM R:PRELIMINARY CALC. FOR INTENSITY; 10E4:CONVERSION FACTOR
884 LET R=(3.1416*D*10E4)/(W*L)
885 REM LOOP TO CALCULATE PATTERN AND PLOT IT
886 FOR X=A TO B STEP .02
887 REM Y:INTENSITY
888 REM 20:SCALE FACTOR FOR PLOT; X:DISTANCE (MM'S)
889 LET Y=20*COS(R*X)*COS(R*X)
890 IF ABS(X)<.0001 THEN 893
891 PRINT TAB(8);INT(X*100+.5)/100;TAB(15); "."; TAB(INT(Y+15.5)); "*"
892 GO TO 895
893 PRINT "...............O....INTENSITY......*"
895 NEXT X
896 LET U=0
897 PRINT
898 RETURN
899 REM
900 REM L INPUT SUBROUTINE
902 PRINT "WHAT IS THE NEW DISTANCE FROM SLITS TO SCREEN (L) ";
903 PRINT "IN METERS";
904 INPUT L
905 REM 1000:  CONVERT L(METERS) TO L(MILLIMETERS)
906 IF 1000*L>= 10*D THEN 912
907 PRINT "THIS DISTANCE IS TOO SMALL FOR GOOD INTERFERENCE PATTERNS."
908 PRINT "TRY ANOTHER VALUE."
910 GO TO 902
912 IF L<=5 THEN 918
913 PRINT "ALTHOUGH ANY DISTANCE LARGER THAN "10*D/1000" METERS"
914 PRINT "IS VALID, ABOVE 5 METERS BECOMES HARD TO SEE.";
915 PRINT "  TRY ANOTHER VALUE."
916 GO TO 902
918 RETURN
919 REM
920 REM D INPUT SUBROUTINE
922 PRINT "WHAT IS THE NEW SLIT SEPARATION (D) IN MILLIMETERS";
924 INPUT D
```

```
926 IF D>=.1 THEN 932
928 PRINT "SLITS ARE SO CLOSE THEY APPROXIMATE A SINGLE SLIT."
929 PRINT "TRY ANOTHER VALUE."
930 GO TO 922
932 IF D<=.1*1000*L THEN 940
933 PRINT "FOR A VALID INTERFERENCE PATTERN, THE SLIT SEPARATION"
934 PRINT "SHOULD BE LESS THAN ".1*1000*L" MILLIMETERS.  TRY";
935 PRINT " ANOTHER VALUE."
938 GO TO 922
940 RETURN
941 REM
942 REM W INPUT SUBROUTINE
944 PRINT "WHAT IS THE NEW WAVELENGTH (W) IN ANGSTROMS";
946 INPUT W
947 IF W>=3000 THEN 954
948 IF W<1000 THEN 959
949 PRINT "A WAVELENGTH OF "W" IS ULTRAVIOLET LIGHT AND NOT VISIBLE."
950 GO TO 956
954 IF W<=8000 THEN 965
955 PRINT "A WAVELENGTH OF "W" IS INFRARED LIGHT AND NOT VISIBLE."
956 PRINT "THE INTERFERENCE PATTERN WILL BE VISIBLE USING DETECTORS"
957 PRINT "ONLY.  TRY ANOTHER WAVELENGTH."
958 GO TO 944
959 PRINT "A WAVELENGTH OF "W" IS X-RAYS AND NOT VISIBLE."
960 GO TO 956
965 RETURN
966 REM
967 PRINT
968 PRINT "*****"
969 PRINT
970 REM MISCELLANEOUS RUNS
972 PRINT "WOULD YOU LIKE A PLOT WITH YOUR OWN VALUES FOR WAVELENGTH"
973 PRINT " (W), SLIT SEPARATION (D), AND DISTANCE FROM SLITS TO"
974 PRINT "SCREEN (L) (1-YES, 0-NO)";
976 INPUT Q9
980 IF Q9<1 THEN 995
982 GOSUB 944
984 GOSUB 922
986 GOSUB 902
988 GOSUB 855
990 PRINT "ANOTHER ONE (1-YES, 0-NO)"
992 INPUT Q8
993 IF Q8>0 THEN 982
994 REM
995 PRINT
996 PRINT "**********"
997 PRINT
998 PRINT "HOPE YOU HAD FUN!"
999 END
```

IV. BASIC RELATIONSHIPS AND ASSUMPTIONS

The model employed in the program and the assumptions which underlie it are listed below.

A. Model

$$Y = \tfrac{1}{2}(1 + \cos RX)$$

where

Y = relative light intensity

X = distance (in centimeters) from center of the viewing screen to the point at which intensity is measured. X is measured along an axis perpendicular to the bright and dark bands on the viewing screen.

$R = 2\pi \cdot 10^{4} \cdot D / W \cdot L$

D = distance (in millimeters) between slits

W = wavelength (in angstroms) of light

L = distance (in meters) between double-slit screen and viewing screen

B. Assumptions

1. The light source is nearly monochromatic. This is not a very strict assumption, since all that is needed is that

$$\frac{\Delta W}{W} << M$$

where

W = central wavelength

ΔW = spread in wavelength about central wavelength

M = number of fringes in the pattern

2. Slit S, in Fig. 1, is narrow enough so that

$$S < \frac{WL_o}{D}$$

3. The slit width in the double-slit screen is less than one tenth the distance between the slits; i.e.,

$$a < 0.1D$$

4. The distance between the double-slit screen and the viewing screen is much larger than the distance between the double slits; i.e.,

$$L >> D$$

5. The double-slit screen and the viewing screen are parallel.

C. Range of Parameters

The program incorporates a series of traps to prevent the use of parameter values outside the range of validity of the model. The ranges of the parameters are:

1. $0.1 < L < 5$ meters
   The lower limit ensures conformance with assumption 4; while the upper one allows for the sharp diminution of intensity with increased distance.

2. $0.1 < D < 0.1L$ millimeters

3. $3000 < W < 8000$ angstroms (slightly wider than the range of visible light)

V. DERIVATION OF MODEL

The general problem of predicting the interference pattern from two slits is very difficult to solve and leads to a very complicated expression for the intensity on the viewing screen. The special problem of two very narrow slits, illuminated by monochromatic light coming from a very narrow slit, and viewed near the axis a long way away is not too difficult. We will derive the intensity expression for the special case, and then, indicate how it changes as some restrictions are lifted, i.e., as the slits get wider, as the illuminating slit gets wider, and as the light source becomes less monochromatic.

The physics of the problem is:

Coherent light has wave-like properties; that is, amplitudes add. Our eyes (and other detectors) detect the square of the amplitude (called the intensity).

A. Location of Bright and Dark Bands

Although there are far more elegant developments available, we shall present a simple geometric one. The geometry is shown in **Figure 1.**

Lines AP and BP are the paths of intersecting light rays passing through slits A and B, respectively. Line AC is perpendicular to line OP (the line from the center point of the double-slit screen to the intersection of the two light rays).

Angles AON and AEO are right angles. Angles OAE and PON are equal (by complementary angles). These angles will be called θ; i.e.,

$$\measuredangle OAE = \measuredangle PON = \theta$$

From triangle PON (a right triangle),

$$\tan \theta = \frac{X}{L}$$

If

$$D << L$$

then

$$\measuredangle EPC << 90^{\circ}$$

and

$$\measuredangle ACB \cong 90^{\circ}$$

Further, triangle ACB is nearly a right triangle, so that



Figure 1. Geometry of light pattern.

$$\sin \theta \cong \frac{Q}{D}$$

If

$$X << L$$

$$\theta \cong \tan \theta = \frac{X}{L}$$

Because ACB is nearly a right triangle, triangles ACB and PON are similar triangles; and, because

$$X << L$$

we see that

$$Q << D$$

therefore

$$\theta \cong \sin \theta \cong \frac{Q}{D}$$

and, finally,

$$\frac{Q}{D} = \frac{X}{L}$$

or

$$X = \frac{QL}{D}$$

If the light rays from A and B are coherent, and if Q = W = wavelength of light from source the rays will reinforce one another, and there will be a bright band at P. Similarly, there will be bright bands at P when

$$Q = 2W,\ 3W,\ 4W,\ ---$$

As a consequence, bright bands are located at

$$X = \frac{KWL}{D}, \text{ for } K = 0, 1, 2, 3, ---$$

When

$$Q = \frac{W}{2}$$

the two waves cancel at P, and we have a dark band. In general, the dark bands are located at

$$X = \frac{JWL}{2D}, \text{ for } J = 1, 3, 5, ---$$

B. Intensity of Light

For this same simple case (coherent, monochromatic illumination, very narrow slits, long distance to the screen), if the amplitude at the screen is written as the sum of two sinusoidal waves, is squared to find the intensity, and is averaged over time to get rid of unobservable variations at twice the light frequency, the actual intensity pattern is

$$I = \frac{I_{max}}{2}\left(1 + \cos\left(2\pi \frac{D}{WL} X\right)\right)$$

The intensity varies from zero to I_{max} and has its maximum value when A is 0, W, 2W, 3W, etc. The relative intensity is defined as

$$Y = \frac{I}{I_{max}}$$

and goes from zero to unity.

$$Y = \frac{I}{I_{max}} = \frac{1}{2}\left[1 + \cos\left(2\pi \frac{D}{WL} X\right)\right]$$

This is the expression used to calculate the relative intensity in the SLITS program.

C. The Effect of the Width of the Double Slits

The ideal fringe pattern is shown in Figure 2. As some of the restrictions on the width of the first slit and the width of the double slits are relaxed the fringe pattern looks more and more like that of Figure 3.

To see how this happens, we draw the more-accurate slit diagram of Figure 4. Now, light travelling along many different paths from each slit can reach the screen. The effect of adding many amplitudes from the two slits of width a (assuming a is still very narrow) is to give a relative intensity of

$$Y = \frac{1}{2}\left[\frac{\sin\left(\pi \frac{a}{WL} X\right)}{\left(\pi \frac{a}{WL} X\right)}\right]^2 \left[1 + \cos 2\pi \frac{D}{WL} X\right]$$

The multiplying factor $\sin^2 \frac{\pi a}{WL} X \Big/ \left(\frac{\pi a}{WL} X\right)^2$ is unity at X = 0 and falls to zero at $X = \frac{WL}{a}$. It is the diffraction pattern of the individual slits and for wide slits, quickly reduces the intensity of all the interference fringes except the central one. The interference fringes for a, but not zero, are shown in Figure 5.

D. Width of the First Slit

If the width of the double slits is kept very small, but the width of of the first slit increased, the light arriving at each of the double slits is the sum of light travelling many paths from different parts of the first



Figure 2. Ideal fringe pattern.



Figure 3. Fringe pattern with finite-width single and double slits.



Figure 4. **Finite-width slits**



Figure 5. Effect of finite-width double slits.



Figure 6. The effect of finite-width single slit.

slit has different (and random!) phases, the intensities and not the amplitudes must be added at the double slits.

The light falling on the double slits is no longer coherent (it no longer all has one phase) and the contrast level of the fringe pattern produced by the double slits is reduced. If all the intensities from the first slit are added, and the double slits are very narrow, the relative intensity in the interference pattern is

$$Y = \frac{1}{2}\left[1 + C \cos\frac{2\pi D}{WL}\right]$$

where C is a function of the width of the first slit, the distance from the first slit to the double slits, the distance between the double slits, and the wavelength. Specifically,

$$C = \left[\frac{\sin\left[\frac{\pi D}{WL_o}S\right]}{\frac{\pi D}{WL_o}S}\right]^2$$

For a very narrow first slit ($S \cong 0$), C = 1 and the relative intensity has maximum value of unity and a minimum value of zero, just as before. As the width of the first slit gets larger, C gets smaller and the fringes have maximum intensity $\frac{1}{2}(1 + C)$ which is less than 1, and minimum intensity $\frac{1}{2}(1 - C)$, which is greater than zero. The fringes now appear as shown in Figure 6. This is a uniform bright background with somewhat brighter and darker lines.

E. The Width Effects Combined

The combined effect of a first slit of width a and double slits of width a leads to the relative intensity pattern

$$Y = \left[\frac{\sin \pi \frac{a}{WL}X}{\pi \frac{a}{WL}X}\right]^2 \left[1 + \left(\frac{\sin \frac{\pi D}{WL_o}S}{\frac{\pi D}{WL_o}S}\right)^2 \cos\frac{2\pi D}{WL}X\right]$$

This is the pattern plotted in Figure 3.

If a laser is used as a light source, the first slit is unnecessary, since all of the light from the laser has the same phase. With laser illumination, C=1, even when the laser light is spread or focussed by a lens. The double-slit interference pattern with a laser as a light source always has maximum contrast, because the laser is a completely coherent source.

If

$$\frac{a}{WL} X << 1$$

and

$$\frac{D}{WL_o} S << 1$$

then

$$\frac{\sin \frac{\pi a}{WL} X}{\pi \frac{aX}{WL} X} \cong 1$$

and

$$\frac{\sin \frac{\pi D}{WL_o} S}{\frac{D}{WL_o} S} \cong 1$$

and the relative intensity simplifies to our original expression; i.e.,

$$Y = \frac{1}{2}\left(1 + \cos \frac{2\pi D}{WL} X\right)$$

F. The Effect of a Spread of Source Wavelengths

If the source is not perfectly monochromatic, the intensities due to light of each wavelength leaving the source add (light of different wavelengths is incoherent) and the total intensity is the sum of many expressions of the form

$$Y = \frac{1}{2}\left(1 + \cos \frac{2\pi D}{WL} X\right)$$

with different values of W. Each color has bright fringes at a different set of values of X (except for X = 0). For instance, if the sources produce light of two wavelengths that differ by 10% (e.g., 6000 Å and 5400 Å) the positions of the bright fringes for these two colors will differ by 10%, but out at the fifth bright fringe this 10% corresponds to a distance of half a fringe width, and the pattern becomes washed out. The number of fringes which can be seen is given by

$$n \cong \frac{1}{2} \frac{W_1}{W_2 - W_1}$$

where W_2 and W_1 are the shortest and largest wavelengths produced by the source. A single bright spectral line from a mercury or sodium source has a spread of wavelengths of about 1 Å so that for the 4358 Å Hg line for instance

$$n \approx \frac{1}{2} \frac{4358}{1} \approx 2000$$

Obviously, it is not hard to make a source that is monochromatic enough to give many more fringes than other limitations of the experiment allow.

G. Experimental Conditions That Ensure a Simple Interference Pattern

The experimental conditions that have to be satisfied in order for the model of the two-slit experiment used on the SLITS program to correspond to an actual experiment are:

1) $\frac{X}{L} \ll 1$; $\frac{D}{L} << 1$

This condition says that the slit spacing and the distance from the center of the screen to the last fringe observed must both be less than the distance from the slits to the screen. This ensures that the distances from the slits to the observation point are nearly equal, and, hence, that the light intensities are equal; and also that the light from the two slits can be approximated as plane waves hitting the screen nearly perpendicularly.

2) $X \ll \frac{WL}{a}$

This says that the distance from the center of the viewing screen to the last fringe observed is much less than the product of the wavelength times the distance to the screen divided by the width of one of the double slits. This ensures that the relative intensity of the brightest part of the fringes will be uniform from the center of the pattern to the last fringe.

3) $D \ll \frac{WL_o}{S}$

The distance between slits must be less than the product of the wavelength times the distance from the slit to the double slits divided by the width of the first slit. This ensures that the light entering the double slits is coherent, and, therefore, that the intensity has the maximum value unity, and the minimum value zero, and the fringes have the maximum possible contrast.

4) $X < \frac{1}{2} \cdot \frac{W_1 L}{D} \cdot \frac{W_1}{W_2 - W_1}$

This says that the distance from the center of the viewing screen to the last fringe should be less than the product of the distance from double slits to screen times the the shortest wavelength squared divided by the slit separation times the difference between the longest and the shortest wavelengths. This ensures that the wavelength spread in the source does not make the fringe maxima of the longest wavelength light fall on the fringe maxima of the shortest wavelength light.

5) Other experimental details are:

a) The first slit and the double slits must be parallel
b) The plane of the double slits and the plane of the viewing screen must be parallel
c) If the distance between the double slits and the viewing screen or between the first slit and the double slits is too great, actual fringe intensity will be very weak (the relative intensity is not affected).